
 3 metodi:

 Comandi CL

 API QWVRCSTK

 Funzione SQL STACK_INFO



Reperire il chiamante in CLP

 In un programma CL si può reperire il chiamante sfruttando 
l'invio/ricezione di un messaggio

dcl        &sender *char 80

dcl        &caller *char stg(*defined) len(10) defvar(&sender 56)

sndpgmmsg msg('Chi mi chiama?') topgmq(*prv (*)) keyvar(&msgkey)

rcvmsg pgmq(*prv (*)) msgtype(*info) msgkey(&msgkey) rmv(*yes) +

sender(&sender)

GETCALLER1.CLLE



API QWVRCSTK/1

 La API QWVRCSTK restituisce lo stack delle chiamate

GETCALLER2.RPGLE

da V5R1

// PROTOTIPO API
dcl-pr RtvStack extpgm('QWVRCSTK');

RecVar char(2000);          // (O) receiver variable
LenRecVar int(10) const;    // (I) receiver variable length
FmtRecVar char(8) const;    // (I) receiver variable format
JobID char(56) const;       // (I) job identification
FmtJobID char(8) const;     // (I) job identification format
Error char(15);             // (I/O) error

end-pr;

// 4º PARAMETRO JobID
dcl-ds JobIDInf;

JIDQName char(26) inz('*');
JIDIntID char(16);
JIDRes3 char(2) inz(*loval);
JIDThreadInd int(10) inz(1);
JIDThread char(8) inz(*loval);

end-ds;

Job corrente

// 1º PARAMETRO RecVar
dcl-ds RecVar len(2000);

BytRtn int(10);
BytAvl int(10);
Entries int(10);
Offset int(10);
EntryCount int(10);
ThreadID char(8);
InfStatus char(1);

end-ds;

https://www.ibm.com/docs/api/v1/content/ssw_ibm_i_75/apis/qwvrcstk.htm


API QWVRCSTK/2

// chiamata API
RtvStack(RecVar :

%size(RecVar) :
'CSTK0100' :
JobIDInf :
'JIDF0100' :
Api_Error);



API QWVRCSTK/3

// voce dello stack
dcl-ds Entry len(256) inz;

EntryLen int(10);          // Length of this call stack entry
ResqtLvl int(10) pos(21);  // Request level
PgmNam char(10) pos(25);   // Program name
PgmLib char(10) pos(35);   // Program library
ModName char(10) pos(49);  // Module name
ModLib char(10) pos(59);   // Module library
CtlBnd char(1) pos(69);    // Control boundary

//  (0 = no control boundary is active,
//   1 = control boundary active)

ActGrpNam char(10) pos(77);// Activation group name
end-ds;

Entry = %subst(RecVar : Offset + 1);
// posizionamento a voce successiva
Offset = Offset + EntryLen;

Nel parametro 
RecVar a partire 
dalla posizione 
specificata in OffSet
viene restituita un 
array con le voci 
dello stack



Funzione SQL STACK_INFO/1

 La funzione di tabella STACK_INFO restituisce lo stack delle 
chiamate di un job

 Occorre l'autorizzazione speciale *JOBCTL se si desidera ottenere 
lo stack di un job di un altro utente

 Occorre l'autorizzazione speciale *SERVICE se si desidera ottenere 
lo stack delle chiamate del LIC o dell'ambiente PASE

da 7.2 TR7

https://www.ibm.com/docs/en/i/7.5?topic=services-stack-info-table-function


Funzione SQL STACK_INFO/2

select ORDINAL_POSITION "Ord.",
PROGRAM_LIBRARY_NAME concat '/' concat PROGRAM_NAME "Programma",
PROCEDURE_NAME "Procedura", STATEMENT_IDENTIFIERS "Istr.",
CONTROL_BOUNDARY "Control boundary",
ACTIVATION_GROUP_NAME "Act.group"

from table(STACK_INFO()) as STACK_INFO
order by ORDINAL_POSITION desc;

job corrente



Funzione SQL STACK_INFO/3

 Metodo 1

GETCALLER3.SQLRPGLE

exec sql
select PROGRAM_NAME
into :PgmName
from table(STACK_INFO()) as STACK_INFO
where PROGRAM_NAME not in('QDBSSUDF2', 'QQQSVRTN', 

'QQQOOOUPCL', 'QDBGETMQO', 'QSQRUN2', 'QSQROUTE', :ps_Program)
order by ORDINAL_POSITION desc
fetch first 1 rows only;

dcl-ds psds PSDS;
ps_Program char(10) pos(1);

end-ds;

escludere dallo stack delle chiamate 
tutti i programmi invocati per 
eseguire la funzione di tabella 

STACK_INFO e il programma in 
esecuzione



Funzione SQL STACK_INFO/3

 Metodo 2

GETCALLER3.SQLRPGLE

exec sql
with
STACK as
(select ORDINAL_POSITION, PROGRAM_NAME

from table(STACK_INFO()) as STACK_INFO
)
select PROGRAM_NAME
into :PgmName
from STACK
where ORDINAL_POSITION <

(select min(ORDINAL_POSITION)
from STACK where PROGRAM_NAME = :ps_Program)

order by ORDINAL_POSITION desc
fetch first 1 rows only;

posizione del 
programma 

corrente

seleziono le voci 
dello stack

precedente al 
programma 

corrente

tabella 
temporanea con lo 

stack completo



Confronto prestazioni

Metodo Durata (ms)

API ~ 4.395.488

funzione SQL ~ 11.670.577 ~ 2-3 volte più lento

Reperimento del chiamante 10.000 volte consecutivamente

I programmi di esempio 
di Power Coffee sono 
sviluppati e testati sui 
server di PUB400.COM

IBM i 7.5
Power 9009-41G
Dispositivo processore: EP52
Ram: 50 Gb

Welcome to PUB400.COM - Your public IBM i 7.5 server

by Holger Scherer (rzkh.de)

https://pub400.com/
https://www.linkedin.com/in/holger-scherer-46a27217/


Riferimenti

 E-mail aziendale: marco.riva@ivolution.it

 Blog: www.markonetools.it

 E-mail blog: info@markonetools.it

 Linkedin: www.linkedin.com/in/marcoriva-mk1

 Twitter: @MarcoRiva73

 Facebook: https://www.facebook.com/markonetools/

 YouTube: youtube.com/@markonetools

13

Power coffee - MK1

mailto:mriva@sirio-is.it
http://www.markonetools.it/
mailto:info@markonetools.it
http://www.linkedin.com/in/marcoriva-mk1
https://twitter.com/MarcoRiva73
https://www.facebook.com/markonetools/
https://youtube.com/@markonetools
http://www.markonetools.it/
http://www.sirio-is.it/
http://www.linkedin.com/in/marco-riva-5975785a


/* ESEMPIO REPERIMENTO CHIAMANTE DA PROGRAMMA CLP */
/* UTILIZZANDO INVIO/RICEZIONE MESSAGGI           */

/* (c) 2022 MarkOneTools - www.markonetools.it */

      pgm

      dcl        &sender *char 80
      dcl        &caller *char stg(*defined) len(10) defvar(&sender 56)
      dcl        &msgkey *char 4

      sndpgmmsg  msg('Chi mi chiama?') topgmq(*prv (*)) keyvar(&msgkey)

      rcvmsg     pgmq(*prv (*)) msgtype(*info) msgkey(&msgkey) rmv(*yes) +
            sender(&sender)

      sndpgmmsg  msg('Il chiamante è ' *cat &caller) tomsgq(*hstlog)

      endpgm



**free
// ESEMPIO REPERIMENTO PROGRAMMA CHIAMANTE CON LA API QWVRCSTK
//
// (c) 2022 MarkOneTools - www.markonetools.it

ctl-opt copyright('MarkOneTools')
        decedit('0,')
        indent(' ')
        option(*nodebugio: *srcstmt: *showcpy)
        expropts(*resdecpos)
        extbinint(*yes);

// prototipo API QWVRCSTK
dcl-pr RtvStack extpgm('QWVRCSTK');
  RecVar char(2000);          // (O) receiver variable
  LenRecVar int(10) const;    // (I) receiver variable length
  FmtRecVar char(8) const;    // (I) receiver variable format
  JobID char(56) const;       // (I) job identification
  FmtJobID char(8) const;     // (I) job identification format
  Error char(15);             // (I/O) error
end-pr;

// DS mappatura 1ยบ parametro API: RecVar (formato CSTK0100)
//  per programmi OPM o ILE
dcl-ds RecVar len(2000);
  BytRtn int(10);     // Bytes returned
  BytAvl int(10);     // Bytes available
  Entries int(10);    // Number of call stack entries for thread
  Offset int(10);     // Offset to call stack entry information
  EntryCount int(10); // Number of call stack entries returned
  ThreadID char(8);   // Returned Thread identifier
  InfStatus char(1);  // Information status
                      //  (' ' no error, I = information not complete, N = no entries returned)
end-ds;

// DS mappatura 4ยบ parametro: job di cui reperire lo stack (corrente) formato JIDF0100
dcl-ds JobIDInf;
  JIDQName char(26) inz('*');    // Job name (10) + User name (10) + job number (6)
                                 //  '*' = the job id which this program is running
  JIDIntID char(16);             // Internal job identifier
  JIDRes3 char(2) inz(*loval);   // reserved
  JIDThreadInd int(10) inz(1);   // Thread indicator
  JIDThread char(8) inz(*loval); // Thread identifier
end-ds;

// DS mappatura 6ยบ parametro API: info errore API
dcl-ds API_Error inz;
  errDSLen int(10) inz(%size(API_Error));
  errDSRtnLen int(10) inz(0);
  errDSMsgID char(7);
  errReserved char(1) inz(x'00');
end-ds;

// DS mappatura singola voce dello stack restituita da API
dcl-ds Entry len(256) inz;
  EntryLen int(10);          // Length of this call stack entry
  ResqtLvl int(10) pos(21);  // Request level
  PgmNam char(10) pos(25);   // Program name
  PgmLib char(10) pos(35);   // Program library
  ModName char(10) pos(49);  // Module name
  ModLib char(10) pos(59);   // Module library
  CtlBnd char(1) pos(69);    // Control boundary
                             //  (0 = no control boundary is active, 1 = control boundary active)
  ActGrpNam char(10) pos(77);// Activation group name
end-ds;

// status program data structure
dcl-ds psds PSDS;
  ps_Program char(10) pos(1);  // current program
end-ds;

dcl-s idx int(5);

// richiamo API per stack chiamate
RtvStack(RecVar :
         %size(RecVar) :
         'CSTK0100' :
         JobIDInf :
         'JIDF0100' :
         Api_Error);
// ciclo sullo stack, individuo il primo programma non di sistema
// quindi diverso da Q* e diverso dall'attuale programma
for idx = 1 to Entries;
  Entry = %subst(RecVar:Offset + 1);
  Offset = Offset + EntryLen;
  if %subst(PgmNam: 1: 1) = 'Q' or
     PgmNam = ps_Program;
    iter;
  endif;
  dsply ('Il chiamante รจ ' + %trim(PgmNam));
  leave;
endfor;

*inlr = *on;


**free

// ESEMPIO REPERIMENTO PROGRAMMA CHIAMANTE CON LA FUNZIONE SQL STACK_INFO
//
// (c) 2022 MarkOneTools - www.markonetools.it

ctl-opt copyright('MarkOneTools')
        decedit('0,')
        indent(' ')
        option(*nodebugio: *srcstmt: *showcpy)
        expropts(*resdecpos)
        extbinint(*yes);

dcl-s PgmName char(10);
dcl-ds psds PSDS;
  ps_Program char(10) pos(1);
end-ds;

// METODO 1
exec sql
  select PROGRAM_NAME
    into :PgmName
    from table(STACK_INFO()) as STACK_INFO
    where PROGRAM_NAME not in('QDBSSUDF2', 'QQQSVRTN', 'QQQOOOUPCL',
                              'QDBGETMQO', 'QSQRUN2', 'QSQROUTE',
                              :ps_Program)
    order by ORDINAL_POSITION desc
    fetch first 1 rows only;

dsply ('1) Il chiamante è ' + %trim(PgmName));

// METODO 2
exec sql
  with
   STACK as
   (select ORDINAL_POSITION, PROGRAM_NAME
      from table(STACK_INFO()) as STACK_INFO
   )
   select PROGRAM_NAME
     into :PgmName
     from STACK
     where ORDINAL_POSITION < (select min(ORDINAL_POSITION)
                                 from STACK
                                 where PROGRAM_NAME = :ps_Program)
     order by ORDINAL_POSITION desc
     fetch first 1 rows only;

dsply ('2) Il chiamante è ' + %trim(PgmName));

*inlr = *on;
return;

